
Verilog — инструмент разработки цифровых электронных схем

Введение
   Язык Verilog был разработан фирмой Gateway Design Automaton как внутренний язык симуляции. Cadence приобрела Gateway в 1989 г. и открыла Verilog для общественного использования. В 1995 г. был определен стандарт языка — Verilog LRM (Language Reference Manual), IEEE1364-1995. Таким образом, датой появления языка Verilog следует считать 1995 г. К этому времени уже успел получить широкое распространение другой язык высокого уровня для описания принципиальных схем — VHDL (Very high-speed IC Hardware Description Language), появившийся еще в 1987 г. Несмотря на похожие названия, Verilog HDL и VHDL — различные языки. В статье сравниваются конструкции этих языков, предполагая, что читателю известен VHDL, но данный материал будет также полезен и для начинающих изучение HDL с языка Verilog. По мнению автора, Verilog — достаточно простой язык, сходный с языком программирования С как по синтаксису, так и по «идеологии». Малое количество служебных слов и простота основных конструкций упрощают изучение и позволяют использовать Verilog в целях обучения. Но в то же время это эффективный и специализированный язык. VHDL обладает большей универсальностью и может быть использован не только для описания моделей цифровых электронных схем, но и для других моделей (например, модели экосистемы). Однако из-за своих расширенных возможностей VHDL проигрывает в эффективности и простоте, то есть на описание одной и той же конструкции в Verilog потребуется в 3–4 раза меньше символов (ASCII), чем в VHDL.
   В Verilog существуют специфические объекты (UDP, Specify-блоки), не имеющие аналогов в VHDL. Также следует упомянуть стандарт PLI (Program Language Interface), который позволяет включать функции, написанные пользователем (например, на С), в код симулятора. В настоящее время важным недостатком Verilog является отсутствие документации, особенно на русском языке. Данная статья является попыткой улучшить эту ситуацию.

Основы синтаксиса

   Для иллюстрации примеров будет использоваться Verilog-XL симулятор от Cadence (www.cadence. com). Это наиболее стандартный, если так можно выразиться, симулятор. Существует несколько проектов бесплатных симуляторов Verilog. Автору известны Veribest и Icarus Verilog. Однако в настоящее время данные продукты не являются завершенными, и их работоспособность на больших проектах, полнота и правильность поддержки стандарта являются неудовлетворительными. Но для примеров, рассмотренных в статье, эти программы вполне подходят. Еще одним неприятным моментом является то, что собрать эти симуляторы для работы под Windows автору не удалось (пакеты поставляются в исходном коде и в скомпилированном виде для ОС Solaris и Linux). Большинство производителей также предлагают бесплатные демонстрационные версии симуляторов Verilog. Неплохой список ссылок на страницы с подобным программным обеспечением находится по адресу http://sal.kachinatech.com/Z/1/index.shtml. Для Windows-платформы весьма популярен ModelSim. Под Linux хорошо зарекомендовал себя finsim , для которого существует и Windows-версия. Также в большинстве современных продуктов, предназначенных для синтеза и верификации проектов на FPGA или ASIC, существует поддержка Verilog HDL и VHDL как для синтеза, так и для симуляции. Симуляторы можно условно разбить на две группы: интерпретирующие и компилирующие в «родной код». Так, Verilog-XL является интерпретирующим симулятором, NC-Verilog — компилирующим, а finsim может работать как в том, так и в другом режиме. Для работы компилирующих симуляторов обычно требуется, чтобы в системе был установлен С-компилятор.
   Запуск симулятора может производиться из командной строки или из графической оболочки, которая формирует строку для запуска. Часть параметров, передаваемая в командной строке, уникальна для конкретного симулятора, но есть основной стандартный набор. Так, вызов симулятора со списком файлов вызовет их исполнение (Verilog Test.V Modul.V). Список файлов можно поместить в отдельный файл и вызывать симулятор с опцией -f (Verilog -f FileList).
   Все приводимые в тексте статьи примеры можно загрузить по адресу http://chat.ru/ ~verilog_intro. Внимание: данное описание не является полным изложением синтаксиса языка; в статье рассмотрены только основные конструкции, необходимые для понимания структуры и принципов языка.

Типы данных

   Verilog поддерживает следующие «стандартные» типы данных: целое — integer (32-битовое со знаком) и real — число с плавающей точкой.
   Для моделирования также используются time (время), специфический тип, применяемый встроенными функциями для моделирования времени, обычно 64-битовое целое; event (событие) — в языке существует ряд операторов и конструкций для работы с событиями.
   Создавать свои типы данных, как в VHDL, нельзя. При разработке синтезируемых моделей из перечисленных типов используется только integer.
   Сигналы, в отличие от VHDL, бывают двух основных типов: «цепи» и «регистры». Самые распространенные из них описываются ключевыми словами wire и reg соответственно. Однако следует помнить, что средство синтеза не всегда реализует reg в виде триггера. Отличие wire от reg состоит в том, что reg способен сохранять присвоенное значение (работает как переменная в языках программирования), а к wire требуется прилагать непрерывное воздействие (driver). То есть wire моделирует провод, который переходит в неопределенное состояние при отключении драйвера.
   Существуют также wand, wor, tri0, tri1, triand, trior и trireg (это цепь, а не регистр!) для моделирования различных типов цепей (wand — wired and, tri0 — резистор к 0, trireg — емкость и т. п.), но такие цепи встречаются редко и потому в статье не рассматриваются.
   Идентификаторы в Verilog являются чувствительными к регистру написания и подчиняются обычным правилам: не могут начинаться с цифры или знака $ и могут содержать буквы, цифры, «$», «_». Существуют так называемые escaped-идентификаторы (в основном в структурных моделях, полученных после синтеза), которые начинаются с «\», содержат любые символы и заканчиваются пробелом или переводом строки.

Пример:

// — это комментарий
/* — и это комментарий
*/
integer i, j, k; // объявление переменных i, j и k типа integer
time start, duration;
real freq_div;
event start_process;
/* далее следуют объявления
однобитовых сигналов */
wire a, b, c;
wire d;
reg store, ff, A; // reg A не совпадает с wire a (case sensitive vs VHDL)
reg \dut/cntr/reg_s ; // escaped идентификатор (пробел перед «;») 

   Для описания шин или регистров неединичной ширины используются диапазоны (range) вида [n:m], где m и n — целые числа или параметры. В языке допускается как m>n, так и наоборот. Но это имеет значение в операциях, для которых важен порядок битов (например, сложение или присвоение целого). Поэтому принято располагать индексы в убывающем порядке:

wire [7:0] data_bus;
reg [3:0] high_nibble, low_nibble; // два 4-битовых регистра 
reg [0:5] a_reg; // регистр с обратным порядком битов не рекомендуется 

   Массивы в Verilog не поддерживаются, но существуют «памяти», собственно одномерный массив или модель памяти:

reg [8:0] Fifo [31:0]; // 32 слова 9-битовой памяти

   В отличие от VHDL в Verilog представление сигналов реализовано в самом языке (а не в библиотеке). Всего существуют четыре типа значений, которые могут принимать сигналы «цепь» и «регистр»: 0, 1, z, x. Первые три соответствуют логическим уровням и состоянию с высоким импедансом. Четвертый (х) означает неопределенное состояние и используется при моделировании неинициализированных сигналов, конфликтов (два выхода с противоположными состояниями соединены вместе), метастабильных состояний триггеров (при нарушении временных соотношений между входами данных и тактовым входом), иными словами, во всех случаях, когда симулятор не может определить значение данного сигнала. В реальном приборе такого сигнала, конечно, не бывает.
   Для записи многоразрядных сигналов (констант) используются следующие конструкции: 1'bz — одноразрядный высокоимпедансный сигнал, 10'd1_000 — десятиразрядное число 1000, записанное в десятичной системе (символ «_» игнорируется), 4'bx01z — четырехразрядный сигнал с неопределенным старшим битом, высокоимпедансным младшим и вторым и третьим, равными 1 и 0 соответственно. То есть запись многоразрядного сигнала представляет собой разрядность, одинарную кавычку «'» (не путать с апострофом «`», использующимся в директивах), основание системы счисления (b, o, d, h) и цифры, использующиеся в данной системе счисления. В двоичной системе допустимо использование символов z и x. Символ подчеркивания служит для облегчения записи и игнорируется. Использование констант без указания разрядности не желательно, так как по умолчанию константа воспринимается с длиной 32 бита. Данные типа integer также могут присваиваться регистрам.

Структурное описание

   Основной структурной единицей Verilog описания является module. Модуль соответствует entity в VHDL. Модуль описывается ключевыми словами module — endmodule. В файле может быть описано несколько модулей. Другие модули могут подключаться к цепям модуля, образуя иерархическую структуру. При запуске Verilog симулятор строит иерархическое дерево из всех модулей, которые обнаружены в файлах, поданных на вход симулятора, и находит модуль верхнего уровня. Если таких модулей несколько, то происходит ошибка. Как правило, модуль содержит список портов — интерфейсных сигналов, которые служат для подключения его в других модулях. Порты бывают трех типов input — входы, output — выходы, inout — двунаправленные. Входы и двунаправленные порты должны иметь тип wire, а выходы могут быть как wire, так и reg. 
   Синтаксис модуля рассмотрим на примере накопительного сумматора:

module NCO (FC, CO, C); // имя модуля и список портов
input FC, C; // входы
output CO; // выход
// описание используемых сигналов
wire [3:0] FC;
wire C; // не обязательно, так как по умолчанию вход является однобитовым проводом
reg [3:0] acc;
reg CO;
// описание поведения системы
initial
begin
acc=0;
CO=0;
end
always @(posedge C) // событие — фронт С
{CO,acc}={CO, acc}+FO;
endmodule 

   Модуль NCO не включает в себя другие модули и является модулем нижнего уровня иерархии. В модуле присутствуют две «поведенческие» конструкции: initial и always. Initial служит для описания действий, которые выполняются один раз (при запуске модели), а always обозначает действия, которые выполняются постоянно. Ключевые слова begin/end имеют такое же значение, как в процедурном языке Паскаль (соответствуют {} в С). Для того чтобы always имело смысл, используется событийный контроль — конструкция @(posedge C), означает «по положительному фронту сигнала С». То есть операция {CO,acc}={CO,acc}+FO выполняется по каждому положительному фронту С. Фигурные скобки обозначают объединение сигналов с различными именами в шину. Объединение может находиться как слева, так и справа от знака «=» в операции присвоения.
   В качестве простейшего модуля верхнего уровня (который можно исполнить на симуляторе) возьмем пример, использующийся при описании процедурных языков:

module hello_word; // интерфейсные порты отсутствуют
initial
$display(«HELLO, WORLD !!!»); // вызов системной функции
endmodule
После запуска симулятора должно получиться приблизительно следующее:
Highest level modules:
hello_word
HELLO, WORLD!!!
0 simulation events 

   В данном примере была использована системная функция $display, которая используется для печати либо форматированной строки (как функция printf из библиотеки stdio в языке C), либо своих аргументов (как writeln в Паскале). Все системные функции и функции, написанные пользователем и подключенные через PLI-интерфейс, начинаются со знака $.
   Немного забегая вперед, следует сказать, что язык Verilog (так же, как и VHDL) изначально предназначался для моделирования и средства синтеза появились позже. Поэтому часть конструкций языка не поддерживается синтезом. Одной из таких конструкций является initial. Для того чтобы осуществить начальную инициализацию, в синтезируемой модели следует предусмотреть специальный сигнал сброса.

module NCO_syn (FC, CO, C, Rst); // имя модуля и список портов
input FC, C, Rst; // входы
output CO; // выход
// описание используемых сигналов
wire [3:0] FC;
wire C, Rst; // не обязательно, так как по умолчанию вход является однобитным проводом
reg [3:0] acc;
reg CO;
// описание поведения системы
always @(posedge C or posedge Rst) // событие — фронт С или Rst
if (Rst)
{CO,acc}=5'b0;
else
{CO,acc}={CO, acc}+FO;
endmodule 

   В данном примере добавлена процедурная конструкция if. Как можно видеть, она подобна конструкции if в языке C, нулевое значение в скобках соответствует false, ненулевое — true. Подробно о процедурных конструкциях будет рассказано позже, но следует заметить, что если в скобках стоит выражение, имеющее после вычисления биты со значением z или х, то выполняется ветвь else.
   Для проверки работоспособности модуля используются испытательные стенды (testbench). Это модуль верхнего уровня, в котором могут использоваться несинтезируемые конструкции (initial) и типы данных (event, time, real). Также испытательные стенды содержат системные функции для вывода информации ($display, $write, $monitor), записи файлов изменения сигналов (vcd — value change dump) для последующего анализа, исследования статистических свойств сигналов и т. п.
   Предположим, что модуль NCO будет использоваться для генерации частот 33 и 40 МГц при тактовой частоте 100 МГц. Для этого в накопительный сумматор на вход FC (frequency code) следует подать 11 и 13 соответственно. При этом полученные частоты будут 34,375 МГц (ошибка 1,375 МГц) и 40,9125 МГц (ошибка 0,9125 МГц), что, предположим, и требуется. В модуле testbench будут использоваться временные задержки — конструкции вида #NN, где NN — время в наносекундах.
   Более подробно различные виды задержек (delay) будут рассмотрены ниже. Оба модуля (NCO и testbench) могут быть записаны либо в один, либо в разные файлы.

`timescale 1ns/10ps //директива симулятора — установка шага времени
(необязательно, так как 1ns/10ps — значения по умолчанию)
module testbench;
// объявление сигналов
reg clk, rst;
reg [3:0] fc1,fc2
wire f1, f2;
// объявление переменных
integer clk_cnt, f1_cnt, f2_cnt;
real ratio;
//построение иерархии
NCO_syn nco1(.Rst(rst), .C(clk), .FC(fc1), .CO(f1)); //подключение по имени
NCO_syn nco2(fc2, f2, clk, rst); //подключение по расположению
initial 
begin
clk=0;
rst=0;
fc1=4'd11;
fc2=4'd13;
clk_cnt=0;
f1_cnt=0;
f2_cnt=0;
end
always #5 clk=~clk; // генератор тактовой частоты 100 МГц
// управление
initial
#1 rst=1'b1; // формирование сброса
#2 rst=1'b0;
#1200; //время симуляции
$display(«toggle: clk «, clk_cnt, « , f1 «, f1_cnt, « , f2», f2_cnt); //вывод результатов
ratio=100.0*$itor(f1_cnt)/$itor(clk_cnt); // $itor — преобразование integer в real
$write(«freqv @ clk=100MHz f1=%f», ratio);
ratio=100.0*$itor(f2_cnt)/$itor(clk_cnt);
$display(« f2=%f», ratio); 
$finish; // завершение симуляции
end
// сбор статистики
always @(posedge clk)
clk_cnt=clk_cnt+1;
always @(posedge f1)
f1_cnt=f1_cnt+1;
always @(posedge f2)
f2_cnt=f2_cnt+1;
//два метода индикации (выбрать один, чтобы не засорять выход)
// 1
// печатает при изменении одного из сигналов
//initial $monitor(«Time %t clk %b rst %b f1 %b f2 %b»,$time,clk,rst,f1,f2);
// 2
//печатает по срезу clk
always @(negedge clk) $write(«Time %t clk %b rst %b f1 %b f2 %b \n», $time, clk, rst, f1, f2);
endmodule. 

   Как можно видеть из приведенного примера, блоков initial и always может быть сколько угодно, исполнение происходит одновременно (о том, как работает симулятор, будет рассказано при рассмотрении операции присваивания). Порядок исполнения различных конструкций определяется только по времени исполнения.
   Построение иерархии (подключение модулей) возможно двумя способами: по имени (указываются имена портов, использованные при описании модуля) или по расположению (порядок сигналов такой же, как в описании модуля).
   После запуска со вторым вариантом печати будет выдано следующее:

Compiling source file «testbench.v»
Compiling source file «nco.v»
Highest level modules:
testbench
Time 1000 clk 0 rst 0 f1 0 f2 0 
Time 2000 clk 0 rst 0 f1 1 f2 1 
Time 3000 clk 0 rst 0 f1 0 f2 0 
Time 4000 clk 0 rst 0 f1 0 f2 1 
Time 5000 clk 0 rst 0 f1 1 f2 0 
Time 6000 clk 0 rst 0 f1 0 f2 0 
Time 7000 clk 0 rst 0 f1 0 f2 1 
Time 8000 clk 0 rst 0 f1 1 f2 0 
Time 9000 clk 0 rst 0 f1 0 f2 1
[…]
Time 118000 clk 0 rst 0 f1 1 f2 1 
Time 119000 clk 0 rst 0 f1 1 f2 0 
Time 120000 clk 0 rst 0 f1 0 f2 1 
toggle: clk 120 , f1 41 , f2 49
freqv @ clk=100MHz f1=34.166667 f2=40.833333
L41 «testbench.v»: $finish at simulation time 120300 

   Время измеряется в единицах, заданных вторым параметром директивы timescale (10 ps), и, соответственно, time 7000 означает 70 нс после старта симуляции. Общее время симуляции составило 1+2+1200=1203 нс. Размерность временных параметров, задаваемых в исходном коде модуля, определяется первым параметром директивы timescale (1 нс).
   Вычисленные значения f1 и f2 отличаются от полученных моделированием. При увеличении длины выборки (времени моделирования) разница будет уменьшаться. 

Verilog — инструмент разработки цифровых электронных схем

Операторы

Синтаксис операторов в языке Verilog подобен синтаксису языка программирования С. К сожалению, отсутствуют операторы ++, -- и все операторы модификации вида (операция)=, например, *=, ^= и т. п. Но в то же время существуют логические операторы, полезные для моделирования цифровых схем: при одинаковом синтаксисе (\, ~|, ^, ~^, & ~&) данные операторы могут быть битовыми (bitwise) и работать с двумя операндами или операторами свертки (reduction) и с одним операндом. Тип выполняемой операции определяется по положению оператора в выражении. Кроме логических операций языком поддерживаются и арифметические операции, но, чтобы получить информацию о синтезируемости выражений с арифметическими операторами, следует ознакомиться с документацией на средства синтеза. Для эффективной реализации (синтеза) арифметических выражений в «железе» возможно потребуется приобретать специальные средства или библиотеки для datapath элементов. 

Рассмотрим применение операторов на примере. 

module op_test;
reg [3:0] D0, D1, D2, D3;
reg [3:0] A, B, C, D; //Verilog case-sensitive
reg a, b, c, d; // А и а — различные переменные
initial
begin
D0=4'b0;
D1=4'b1111;
D2=4'b1010;
D3=4'b01xz;
A=D1~^D2; // bitwise операция — два операнда
a=~^D2; // reduction операция — один операнд
B=D0^D3;
b=^D3;
C=D2&D3;
c=&D3;
D=D2|D3;
d=|D3;
$display(«A=%b a=%b B=%b b=%b C=%b c=%b D=%b d=%d», A, a, B, b, C, c, D, d);
end
endmodule

Полученный результат будет выглядеть так: 

Highest level modules:
op_test
A=1010 a=1 B=01xx b=x C=00x0 c=0 D=111x d=1

При выполнении логических и битовых операций состояние с высоким импедансом (z) воспринимается как неопределенное (x). При моделировании комбинаторная логика обычно способствует распространению x, но если, например, на один из входов элемента И (And) подан 0, то, независимо от значения на других входах, на выходе будет 0. Это иллюстрируют полученные в примере значения С и с. 

В языке присутствует условный оператор ?:, который работает так же, как и в языке С. Таким образом, простейшей записью мультиплексора из 2 в 1 является: 

assign Y=(SEL)?A:B;

Список операторов языка Verilog: 

{} concatenation
+ — * / arithmetic
% modulus
> >= < <= relational
! logical negation
&& logical and
|| logical or
== logical equality
!= logical inequality
=== case equality *
!== case inequality *
~ bit-wise negation
& bit-wise and
| bit-wise inclusive or
^ bit-wise exclusive or
^~ or ~^ bit-wise equivalence
& reduction and
~& reduction nand
| reduction or
~| reduction nor
^ reduction xor
~^ or ^~ reduction xnor
<< shift left
>> shift right
<<< arithmetic shift left
>>> arithmetic shift right
?: conditional

Операторы, помеченные *, будут рассмотрены при описании поведенческих конструкций. 

Часть операторов повторяется. Например, << и <<< выполняют одинаковое действие, а последовательное выполнение ~ и ^ имеет такое же значение, как и ^~ или ~^. 

Применяться операторы могут как к цепям (wire), так и к сигналам (reg) и переменным. При этом используются различные типы присвоения. Для цепей, которые являются моделью физического соединения (провода), требуется подключение непрерывного воздействия, которое моделируется непрерывным (continuous) присвоением. Значения же регистров и переменных могут изменяться в результате процедурных действий и сохраняться между воздействиями (так же, как и переменные процедурного языка программирования). Для моделирования этого используется процедурное (procedural) присвоение. 

Непрерывное присвоение употребляется вне процедурных блоков initial или begin и используется либо в описании цепи, либо с ключевым словом assign (существуют также процедурные непрерывные присвоения в блоках initial или always с ключевыми словами assign и deassign). Слева от оператора непрерывного присвоения (=) должен находиться объект типа цепь. При изменении значения какого-либо из объектов, входящих в выражение справа от =, данное выражение будет вычислено, и новое значение будет присвоено. 

Ниже приведен пример, иллюстрирующий работу непрерывного присвоения. 

module assign_test;
reg [3:0] data;
wire parity, forth;
wire gnd=1'b0; //обьявление присвоение
wire y=(data[0])?data[1]:gnd; //мультиплексор
assign parity=^data; //непрерывное присвоение
assign forth=~|data[1:0]; //еще одно присвоение
initial
$monitor(data,, parity,, forth,, y); //запускаем системную 
функцию для индикации изменений
initial
for (data=0; data!=15; data=data+1'd1) //переберем варианты
#1; //задержка нужна, чтобы $monitor работал правильно
endmodule

Результат работы этого примера: 

Highest level modules:
assign_test
0 0 1 0
1 1 0 0
2 1 0 0
3 0 0 1
4 1 1 0
5 0 0 0
6 0 0 0
7 1 0 1
8 1 1 0
9 0 0 0
10 0 0 0
11 1 0 1
12 0 1 0
13 1 0 0
14 1 0 0
15 0 0 1

Процедурные присвоения бывают двух типов: blocking (=) и nonblocking (<=). В большинстве случаев разработчики переводят эти названия на русский язык как «блочные» и «не блочные» присвоения, но по смыслу более подходящим кажутся термины «блокирующие» и «не блокирующие». Для того чтобы понять разницу между данными типами присвоения, нужно рассмотреть работу Verilog симулятора. В реальном объекте (цифровой схеме), который моделируется с помощью языка Verilog, события могут происходить одновременно: при изменении входного сигнала во всех элементах начинаются процессы, которые протекают одновременно и приводят к каким-либо изменениям выходных сигналов. Программа симулятора не может обрабатывать события одновременно и создает списки событий, которые должны выполняться последовательно. Когда все события из списка выполнены, симулятор переходит к следующему временному шагу: увеличивает значение текущего момента времени на некоторый временной интервал (второй параметр директивы `timescale) и выполняет список событий, которые должны произойти на этом шаге. Об исполнении событий в других временных шагах (механизме задержек) будет рассказано далее, а сейчас рассмотрим события, происходящие «одновременно» — в одном временном шаге. 

Допустим, имеется следующее описание: 

always @ (posedge CLK) a=b;
always @ (posedge CLK) b=a;

Пусть а и b — регистры единичной длины, и к моменту положительного фронта тактового сигнала CLK а = 0, b = 1. Какое значение будут иметь эти переменные после прохождения фронта? Это не определено в языке и зависит от того, в какой последовательности операции присваивания попадут в список. То есть получается конструкция, поведение которой непредсказуемо. Это значит, что либо обе эти переменные будут равны 0, либо 1. Конкретный результат зависит от симулятора и от последовательности, в которой будут обработаны строки исходного файла. «Блокирующее», или «блочное», присвоение (=) блокирует исполнение других операций в блоке последовательных операций (до тех пор, пока не будет выполнена данная операция). Использование «блокирующего» присвоения в подобной конструкции в конкурентно исполняемых блоках нежелательно, и его следует избегать. Но в то же время, если в блоке требуется провести последовательное исполнение операторов, следует применять данный тип присвоения. 

Конструкция 

always @ (posedge CLK)
begin
a=0;
b=a;
end

гарантирует, что после фронта CLK обе переменные — a и b — будут обнулены. 

Если же предыдущие примеры переписать с использованием «неблокирующего», или «неблочного», присвоения (<=), то поведение модели изменится: 

always @ (posedge CLK) a<=b;
always @ (posedge CLK) b<=a;

В этом случае в список событий, исполняемых во временном шаге после изменения CLK, эти операции будут помещены параллельно, то есть переменные обменяются своими значениями. После прохождения фронта CLK переменные изменят свои значения: а = 1, b = 0. Данная конструкция будет эквивалентна следующей: 

always @ (posedge CLK)
begin
a<=b;
b<=a;
end

и описывает два D-триггера, выход каждого подан на вход другого, а на тактовые входы подан сигнал CLK. При этом последовательность записи: a<=b; b<=a или b<=a; a<=b — не играет роли, так как моделируются одновременно происходящие события. 

Вторая конструкция в этом случае: 

always @ (posedge CLK)
begin
a<=0;
b<=a;
end

описывает два D-триггера. На вход одного (а) подан 0, а выход подключен к входу другого (b). На тактовые входы подан сигнал CLK. 

Если дополнить первый триггер сигналом асинхронной установки, получится схема, которая реально может быть использована для синхронизации и нормирования коротких импульсов. 

always @ (posedge CLK or posedge SET)
begin
if (SET) a<=1'b1;
else
begin
a<=0;
b<=a;
end
end

Проверим работоспособность данной схемы следующим испытательным стендом (это упрощенный testbench — правильный пример создания испытательных стендов приведен в первой части статьи): 

`timescale 1ns/10ps
module test;
reg CLK, SET, a, b;
always
#5 CLK=~CLK; //100 МГц тактовый сигнал
always @ (posedge CLK or posedge SET) //испытываемый блок
begin
if (SET) a<=1'b1;
else
begin
a<=0;
b<=a;
end
end
initial //инициализируем переменные
begin
CLK=0;
SET=0;
a=0; b=0;
$monitor($time,,SET,,a,,b);
end
initial #100 $finish; // через 100 нс завершить моделирование
initial //подача «асинхронного» сигнала SET
begin
#57 SET=1'b1;
#1 SET=1'b0;
end
endmodule

Результат работы этого теста: 

Highest level modules:
test
0 0 0 0
57 1 1 0
58 0 1 0
65 0 0 1
75 0 0 0

показывает, что асинхронный сигнал SET, длительностью 1 нс (фронт на 57 нс после пуска), был синхронизирован по фронту CLK в регистре b — длительность 10 нс (фронт 65 нс, срез 75 нс). 

Для пользователей VHDL можно провести параллель между variable assignment (:= VHDL) и blocking assignment (= Verilog), signal assignment (<= VHDL) и nonblocking assignment (<= Verilog) соответственно. Но следует учесть, что в процедурных конструкциях Verilog различий между регистром и переменной не делается. 

Временной и событийный контроль

Так как Verilog используется для моделирования физических систем, то большое внимание уделено привязке события к определенному моменту времени. Для этого используется механизм задержек. 

Следует заметить, что средства синтеза (как для Verilog, так и для VHDL) игнорируют временные конструкции в исходном коде. Для правильной генерации цифровой схемы (нет листа) из библиотечных элементов в средства синтеза, наряду с описанием на языке высокого уровня вводятся файлы, содержащие описания «constrain». В этих файлах описываются временные ограничения распространения сигналов. Применяемый для этого синтаксис не стандартизован и определяется используемым средством синтеза. Профессиональная работа с HDL-языками высокого уровня подразумевает не только умение создавать грамотные поведенческие описания, но и умение правильно управлять средством синтеза с помощью «constrain». 

Однако для моделирования временной контроль необходим, и в примерах этой статьи неоднократно использовались выражения вида #<число>. Вместо числа может использоваться выражение, содержащее целые и/или вещественные параметры. В синтаксисе языка определено несколько форм временных задержек для описания различных технологических условий, наиболее полной из которых является так называемая mintypmax форма — #(min, typ, max), например #(2, 3, 4). Но данные конструкции используются, как правило, для моделирования на уровне вентилей (нет листа), а более эффективным методом таких описаний является SDF-аннотация с применением специальных SDF-файлов. Поэтому в статье будет использоваться упрощенная форма задания задержки с одним параметром #typ. Для анализа исходных файлов «третьей стороны» можно считать, что всегда используется типовое (среднее) значение. 

Рассмотрим применение задержки в непрерывном присвоении. 

assign #10 c=a^b;

Данная конструкция описывает элемент «исключающее или» с задержкой распространения 10 нс (вернее, 10 единиц первого параметра директивы `timescale, который по умолчанию равен 1 нс). Все задержки в непрерывных присвоениях являются инерциальными. Это означает, что если сигнал А изменит свое состояние на время, меньшее 10 нс, а затем изменит еще раз, то изменения сигнала С не произойдет. Для того чтобы произошло изменение сигнала С требуется, чтобы сигнал А был зафиксирован в новом состоянии более 10 нс. Такая модель поведения соответствует распространению задержки при прохождении через элементы электронной схемы. Другой тип задержки — транспортная задержка, которая обеспечивает точное соответствие формы задержанного сигнала и сдвиг его по шкале времени. В отличие от VHDL (в котором существуют спецификаторы типа задержки inertial и transport) Verilog не позволяет использовать оба типа задержек в одном моделировании. Для переключения типа задержки служит ключ командной строки запуска Verilog-симулятора. 

Рассмотрим задержки в процедурных блоках. Первая форма — простая задержка (она использовалась в примерах статьи). Она имеет вид: 

#1 x=y;

Такая задержка вызывает остановку последовательного блока на 1 нс (не влияет на обработку конкурентно исполняемых блоков). Она не обязательно используется с оператором присваивания, может использоваться с пустым оператором #1. То есть #1 x=y может быть записано в такой форме #1; x=y или #0.5; #0.5 x=y. Поведение этих конструкций одинаково. 

Также в процедурном блоке задержка может встречаться с другой стороны знака =. Это так называемая intra-assignment delay. 

x=#1 y;

В этом случае вначале происходит вычисление выражения, затем задержка, а затем присвоение. То есть рассматриваемый пример эквивалентен 

tmp=y;
#1; x=tmp;

Кроме временного контроля существует событийный контроль (который можно считать другой формой временного контроля). Признаком событийного контроля является знак @. В рассмотренных ранее примерах событийный контроль использовался в блоках always. Отличие Verilog от VHDL в данном случае состоит в том, что для описания фронтов и срезов сигналов используются не специализированные атрибуты сигнала, а специальная конструкция языка. Это создает впечатление, что разработчики языка Verilog несколько лучше представляли себе процесс разработки цифровых схем. Событийный контроль используется в процедурных блоках так же, как и временной контроль. При этом задержка исполнения происходит не на фиксированный временной интервал, а до тех пор, пока не произойдет нужное событие. 

События бывают следующих типов: 

· @(name) — изменение name, при этом name может являться цепью, регистром, переменной или переменной типа event; 

· @(posedge A) или @(negedge А) — фронт или срез сигнала А, при этом А — однобитовый регистр или цепь; 

· комбинация перечисленных событий с ключевым словом «or», например @(posedge CLK or posedge SET); в этом случае следует различать «or» с одноименной логической операцией; в событийном контроле «or» означает, что ожидается любое из перечисленных событий, а не определенный результат логической операции.

Для генерации синхронного сброса может использоваться такая конструкция: 

nReset=0;
repeat (16) @(posedge CLK);
nReset=1'b1;

Конструкция во второй строке примера обеспечивает задержку на 16 тактов, вернее, задержку на прохождение 16 фронтов сигнала CLK. 

(Продолжение следует) 

Временной и событийный контроль

Завершая рассмотрение временного и событийного контроля, следует упомянуть о применении intra-assignment delay в «неблокирующем» присвоении, то есть в конструкциях вида: 

x<=#1 y; 
a<= @(posedge c) b;

Поведение этих конструкций таково, что значение выражения вычисляется и блокирования последовательного исполнения операций не происходит, но новое значение будет присвоено только по истечении времени, указанного во временной конструкции, или после совершения события, указанного в событийной конструкции. В терминах работы программы Verilog симулятора операция присвоения переносится на другой временной шаг. В работе таких конструкций проявляется интересное отличие Verilog симуляторов от VHDL-симуляторов. В VHDL каждая следующая по тексту программы операция присвоения одному и тому же сигналу отменяет предыдущую, даже если ее исполнение должно произойти в более ранний момент времени. В Verilog все подобные операции будут помещены в список для соответствующего временного шага, и сигнал, изменение которого вызывают эти операции, будет изменяться в соответствии со всеми операциями. Какой механизм поведения более правильный — вопрос спорный. Так как при синтезе временной контроль игнорируется, да и непонятно, каким образом должна синтезироваться конструкция с присвоением из нескольких источников без специальной разрешающей функции, то это отличие может проявляться только на уровне моделирования с несинтезируемыми элементами. В то же время для работы с несинтезируемыми элементами Verilog предлагает операции, способные отменить (вернее, «пересилить») все остальные операции присвоения к одному определенному сигналу. Эти операции присвоения записываются с ключевыми словами force и release. 

Проиллюстрировать поведение можно следующим примером: 

/* это пример на VHDL 
LIBRARY ieee; 
USE ieee.STD_logic_1164.all; 
USE ieee.std_logic_arith.all; 
ENTITY AT91R IS 
PORT(NRD: OUT std_logic:='0'); 
END AT91R; 
ARCHITECTURE EBI OF AT91R IS 
BEGIN 
modeler: PROCESS 
BEGIN 
NRD <= '1' AFTER 100 ns; 
WAIT FOR 30 ns; 
NRD <= 'Z' AFTER 30 ns; 
WAIT; 
END PROCESS modeler; 
END EBI; 
*/

В результате через 60 нс от начала симуляциивыход переходит в Z-состояние и далее не изменяется. Таким образом, можно видеть, что последующая операция отменяет предыдущую. 

Если переписать этот модуль на Verilog без учета изложенного выше, получим (обратите внимание на лаконичность языка Verilog): 

module AT91R (nrd); 
output nrd; 
reg ndr; 
initial 
begin: modeler //пример именованного блока 
ndr<= #100 1'b1; 
#30; 
ndr<= #30 1'bz; 
end 
endmodule 
При этом поведение будет другим. Написав соответствующий testbench, можно увидеть: 
Highest level modules: 
tst 
0 x 
60 z 
100 1

Это значит, что на 60 нс сработает ndr<= #30 1'bz, а затем на 100 нс — расположенный выше по тексту ndr<= #100 1'b1. В Verilog предыдущая операция не отменяется. Для того чтобы поведение модуля было таким же, как и поведение кода VHDL, нужно записать его следующим образом: 

module AT91R (nrd); 
output nrd; 
reg ndr; 
initial 
begin 
ndr<= #100 1'b1; 
#30; 
#30 force ndr = 1'bz; 
end 
endmodule

Использовать force c «неблокирующим» присвоением и/или intra-assignment delay нельзя. Для того чтобы разрешить дальнейшее использование присвоения в других параллельных блоках, сигнал должен быть отпущен с помощью release (например, release ndr;).Завершая обзор временного контроля, следует упомянуть еще об одной форме задержки — нулевой задержке. В коде Verilog встречаются такие конструкции: #0 a=b. Нулевая задержка означает, что операция будет выполнена в самом конце текущего временного шага. Если в одном временном шаге встречается несколько нулевых задержек, то между собой их порядок не определен. 

Поведенческие конструкции

В поведенческих блоках initial или always могут применяться конструкции управления, сходные с операторами процедурных языков. Данные поведенческие конструкции подразделяются на несколько групп:  

группа принятия решений: if-else-if, case, casez, casex; 

группа повторений: repeat, while, for, forever; 

группа параллельного исполнения: fork-join; 

оператор wait;
Конструкция if записывается следующим образом: 
if ( ) 

else 


Для выбора из нескольких вариантов могут применяться вложенные if. 

if ( ) 

else if ( ) 

else if ( ) 

else 


Здесь expression — любое выражение языка, а statement — оператор или группа операторов между begin и end. Ветвь else может отсутствовать, но если имеются вложенные if (как в примере), то else относится к ближайшему if. Для изменения порядка следует пользоваться командами begin и end. Если получаемое в выражении expression значение не равно 0 и не является неопределенной (x или z), то выполняется ветвь statement1, иначе — statement2. Следует помнить, что так же, как в языке С, операция сравнения записывается как == (два знака «=»), в отличие от операции присваивания = (один знак). Но операции сравнения при неопределенных операндах возвращают неопределенное значение (x). Поэтому в поведенческом моделировании (не принимается средствами синтеза) могут использоваться операции === (три знака =) и !==. Эти операции позволяют произвести литеральное сравнение определенных битов в выражении. Еще раз обратим внимание на то, что выражение expression не является выражением какого-либо специального типа (boolean), а является любым выражением, которое может быть приведено к типу integer. Здесь прослеживается аналогия с языком С, единственное отличие от которого состоит в том, что Verilog integer может принимать неопределенные значения (x или z). В этом случае выполняется ветвь else. 

Исполнение такого кода: 

module if_test; 
initial 
begin 
if (2*5) $display(«2*5!= 0 ==> true»); 
if (2*0) $display(«never print this»); 
else $display(«2*0!= 0 ==> false»); 
if (1'bz) $display(«never print this»); 
else $display(«undefined ==> false»); 
if (1'bx) $display(«never print this»); 
else $display(«undefined ==> false»); 
end 
endmodule

даст следующее: 

Highest level modules: 
if_test 
2*5!= 0 ==> true 
2*0!= 0 ==> false 
undefined ==> false 
undefined ==> false

Следующий пример иллюстрирует применение операторов сравнения (сколько = в каком случае ставить). 

module if_test; 
reg a,b,c,d; 
initial 
begin 
a=(2'b10>3'b001);b=(2'b10==3'b001);c=(2'b10>2'b0x);d=(2'b10==2'bz0); 
$display(«a=%b b=%b c=%b d=%b»,a,b,c,d); 
a=(2'b10!==3'b01);b=(2'b10===2'b10);c=(2'b10!==2'b0x);d=(2'bx1===2'bz1); 
$display(«a=%b b=%b c=%b d=%b»,a,b,c,d); 
a=(2'b0x==2'b0x);b=(2'bx1!=2'bx1);c=(2'b0x===2'b0x);d=(2'bx1!==2'bx1); 
$display(«a=%b b=%b c=%b d=%b»,a,b,c,d); 
end 
endmodule 
Результат: 
a=1 b=0 c=x d=x 
a=1 b=1 c=1 d=0 
a=x b=x c=1 d=0

Для выбора из нескольких вариантов также применяется оператор case. Например, данная конструкция реализует дешифратор, подобный К155ИД3: 

case (rega) 
4'd0: result = 10'b0111111111; 
4'd1: result = 10'b1011111111; 
4'd2: result = 10'b1101111111; 
4'd3: result = 10'b1110111111; 
4'd4: result = 10'b1111011111; 
4'd5: result = 10'b1111101111; 
4'd6: result = 10'b1111110111; 
4'd7: result = 10'b1111111011; 
4'd8: result = 10'b1111111101; 
4'd9: result = 10'b1111111110; 
default result = 'bx; 
endcase

Оператор case является «непроваливающимся», в отличие от оператора switch языка С, и гарантирует исполнение одной ветви. В случае если ни одно из условий не совпадает, выполняется ветвь default. Допустимо другое применение — наоборот, в case — константа, а в ветвях — вычисляемые выражения, либо переменные находятся и там, и там (такого использования оператора выбора в процедурных языках, как правило, нет). Оператор case часто используется в синтезируемом коде для синтеза FSM и мультиплексоров. При этом в несинтезируемых моделях (а в некоторых средствах синтеза и в синтезируемых) в выражениях case могут использоваться литералы с неопределенными значениями. Для поведенческого моделирования используются операторы casez и casex, которые особым образом обрабатывают не-определенные состояния. Синтаксис casez и casex подобен синтаксису case. При этом добавляется символ «?», используемый в двоичной записи литерала для того, чтобы замаскировать биты, которые не должны влиять на принятие решения.

Для демонстрации обработки неопределенных состояний операторами case, casez и casex рассмотрим следующий пример: 

module case_test; 
integer a,b,c,d; 
reg clk; 
always #5 clk=~clk; 
always 
begin: demo 
integer i; 
for(i=0; i<16; i=i+1) 
begin 
$write(«i = %d «,i); 
casex (i) // x, z,? — the same function — ignore bit 
4'b0xxx: $display («less than 8»); 
4'b10zz: $display («not less than 8 and less than 12»); 
4'b11?1: $display («not less than 12 and odd»); 
4'bx?z?: $display («other case»); 
default $display («never print this»); 
endcase 
wait (clk==1'b1); // the same as @(posedge clk) 
end 
end // demo 
initial clk=0; 
endmodule

Данный пример не содержит $finish, а события будут происходить непрерывно из-за блоков always. Поэтому он будет исполняться «вечно». Чтобы его остановить, нужно воспользоваться средствами среды.

Приведем фрагмент результата работы: 

i = 0 less than 8 
i = 1 less than 8 
i = 2 less than 8 
i = 3 less than 8 
i = 4 less than 8 
i = 5 less than 8 
i = 6 less than 8 
i = 7 less than 8 
i = 8 not less than 8 and less than 12 
i = 9 not less than 8 and less than 12 
i = 10 not less than 8 and less than 12 
i = 11 not less than 8 and less than 12 
i = 12 other case 
i = 13 not less than 12 and odd 
i = 14 other case 
i = 15 not less than 12 and odd 
i = 0 less than 8

Таким образом, case проверяет литеральные совпадения, его можно сравнить с использованием if и === (= три раза). В операторах casex и casez биты со значениями x и z или только z игнорируются. В литералах сравнения x и z (casex) или z (casez) могут быть заменены «?».

Операторы повторения могут встречаться в синтезируемом коде. При этом запись упрощается и становится более понятной. В испытательных стендах и несинтезируемых моделях использование операторов повторения имеет такой же смысл, как и в процедурных языках программирования. Операторы for и repeat были продемонстрированы ранее. Так как Verilog не позволяет воспользоваться вечным циклом языка С (for(;;)), то введен оператор forever. Для выхода из циклов (блоки должны быть именованы) служит оператор disable. Продолжая сравнение с языком С, отметим, что disable работает как операторы С break и continue. 

initial 
begin:break 
for(i = 0;i 
begin:continue 
@clk 
if(a == 0) // «continue» loop 
disable continue; 
... ... 
@clk 
if(a == b) // «break» from loop 
disable break; 
... ... 
end 
end

Еще один цикл while имеет следующую форму: 

while (condition) 
begin 
statement 
step_assignment; 
end

Операторы циклов взаимозаменяемы, и выбор определяется личными предпочтениями программиста.

Цикл repeat может использоваться в intra-assignment delay для описания задержки в несколько циклов. Например, таким образом: 

a = repeat(3)@(posedge clk) b.

Операторы fork-join служат для параллельного исполнения ветвей кода в одном процедурном блоке. Это является несинтезируемой конструкцией и используется редко.Оператор wait (см. пример casex) используется для приостановки конкурентно исполняемого блока до тех пор, пока не будет выполнено его условие (как правило, элементы выражения условия wait должны изменяться в другом блоке). 

Процедуры и функции

Функции применяются, как правило, для моделирования комбинаторной логики, которую средства синтеза генерируют по описанию функции. В функциях запрещен временной контроль. При вызове функции создается регистр, размерность и имя которого совпадают с размерностью и именем функции. Через этот регистр функция возвращает результат своей работы. Функция может возвращать также целое или вещественное число. Все параметры, передаваемые в функцию, имеют тип input. 

function [7:0] swap; 
input [7:0] byte; 
begin 
swap = {byte[3:0],byte[7:4]}; 
end 
endfunction

Вызов функции осуществляется следующим образом: 

a=swap(b);

Прежде чем рассматривать процедуры, рассмотрим иерархическую структуру Verilog-модели и области видимости объектов. Локальные переменные (сигналы или параметры) могут объявляться внутри модуля, именованного блока, процедуры или функции. Если в локальной области видимости симулятору не удается обнаружить переменную, то поиск продолжается в более «высокой» области видимости до тех пор, пока не дойдет дело до переменных, сигналов, параметров, объявленных внутри модуля. Если внутри модуля переменная не обнаружена, то выдается сообщение об ошибке. Таким образом, модуль является высшим элементом иерархии областей видимости. Для доступа к объектам в других модулях, собранных в иерархическую структуру, служит операция разрешения контекста. Эта операция задается с помощью имен модулей или именованных блоков внутри модуля (процедур или функций), разделенных точкой. Так, в примере с делителем частоты (он описан в первой части статьи), для доступа к внутреннему регистру acc модуля NCO_syn из модуля верхнего уровня testbench нужно воспользоваться такой конструкцией: <имя включения (instance)>.acc. 

// 2 
always @(negedge clk) $write(«Time %t clk %b rst %b f1 %b f2 %b phase1 %b phase2 %b\n», $time, clk, rst, f1, f2, nco1.acc, nco2.acc );

При этом на печать будут выдаваться значения фазы (из области видимости модулей NCO_syn) 

Time 112000 clk 0 rst 0 f1 1 f2 1 phase1 0000 phase2 0000 
Time 113000 clk 0 rst 0 f1 1 f2 1 phase1 1011 phase2 1101 
Time 114000 clk 0 rst 0 f1 0 f2 0 phase1 0110 phase2 1010 
Time 115000 clk 0 rst 0 f1 1 f2 1 phase1 0001 phase2 0111 
Time 116000 clk 0 rst 0 f1 1 f2 0 phase1 1100 phase2 0100 
Time 117000 clk 0 rst 0 f1 0 f2 1 phase1 0111 phase2 0001 
Time 118000 clk 0 rst 0 f1 1 f2 1 phase1 0010 phase2 1110

Также можно двигаться вверх или вниз по иерархии включения, используя абсолютные (начинающиеся с модуля высшего уровня) либо относительные имена, что очень похоже на методы работы с файловой системой. 

Такой механизм доступа создает ограничение, налагаемое на локальные переменные. Также следует вспомнить о параллельном исполнении и о том, что процедуры в языках HDL не «вызываются», а «разрешаются». Такое название свидетельствует о том, что одновременно может исполняться несколько копий, так как в процедуре разрешен временной и/или событийный контроль. Но при этом память для локальных переменных не выделяется, то есть копии одной и той же процедуры, работающие параллельно, будут «портить» друг другу локальные переменные. Иллюстрацией является практическая невозможность рекурсивных процедур или функций. С этим может столкнуться программист, работавший ранее с процедурными языками.

Синтаксис процедур следующий: 

task my_task; 
input a, b; 
inout c; 
output d, e; 
reg foo1, foo2, foo3; 
begin 
// the set of statements that performs the work of the task 
c = foo1; // the assignments that initialize 
d = foo2; // the results variables 
e = foo3; 
end 
endtask

При этом любое число параметров может передаваться в/из процедуры. Вызов (или «разрешение») процедуры производится следующим образом: 

my_task (v, w, x, y, z);

Функция не имеет права вызывать процедуру, а процедура может разрешать другие процедуры и вызывать себя рекурсивно (см. локальные переменные) или вызывать функции. 

Системные функции

В примерах данной статьи неоднократно использовались системные функции $monitor, $display, $write, $finish, $time. Это малая часть средств, которые предоставляются Verilog-системой программисту для анализа результатов моделирования. Благодаря наличию механизма PLI, обеспечивающего подключение исполняемого кода (написанного либо пользователем, либо третьей стороной), число системных функций и задач, которые выполняются с их помощью, очень велико. Основное назначение системной функции — сбор/анализ информации и взаимодействие с системой. Признак системной функции — $. 

Остановимся на наиболее популярных системных функциях: 
$finish — завершение моделирования; 
$stop — переход в интерактивный режим; 
$display, $write — вывод данных в stdout (дублируется в лог-файл); ведет себя либо как C-функция printf с формат-строкой (поддерживаются дополнительные форматы, например %b — бинарный), либо как процедура write языка Pascal с разделенными запятой аргументами; $display завершает вывод «переводом строки»; 
$monitor — отслеживает изменения аргументов, в конце каждого временного шага печатает при обнаружении изменения значения; формат как у $display; 
$readmemb, $readmemh — обеспечивают считывание данных (в бинарном или шестна-дцатеричном представлении) из файла в память (см. первую часть статьи); формат файла очень простой: в каждой строке либо слово требуемой разрядности, либо конструкция @<адрес загрузки>; Очень удобно применять для моделирования ПЗУ; 
$system — выполняет команду ОС (вызов С функции system()); 
$fopen, $fclose, $fwrite, $fmonitor — файловые операции, позволяют производить запись в файлы; 
$dumpfile, $dumpvars — запись дамп-файлов позволяют записать изменения сигналов модуля, всего проекта или отдельных в специальном формате для последующего изучения; очень полезный и сильный механизм; 
$time — возвращает время симуляции; 
$itor, $random — численные функции, выполняют преобразования или возвращают результат математической функции. 

Это малая часть стандартных функций. Полный список следует искать в документации к симулятору. Также есть функции, которые не являются стандартными (в настоящий момент времени), но поставляются в виде отдельных объектных модулей или С кода. Примером таких функций является $utConnectivity, записывающая список соединений модели для последующего просмотра с помощью программы Undertow ( http://www.veritools.com/), или $togle_count, служащая для сбора статистики переключения сигналов. 

Продолжение следует 

Verilog — инструмент разработки цифровых электронных схем

Сергей Емец
Параметры

При описании цифровых схем возникает необходимость задания констант, определяющих какие-либо фиксированные параметры. Эти константы могут описывать характерные задержки, ширину шин или любой другой параметр, не изменяющийся во время симулирования модели и известный на момент компиляции. Но в то же время при использовании модели одного и того же модуля в различных технологических условиях или включении его различным образом в модули высшего уровня данные константы должны иметь возможность изменяться. В языке VHDL данную функцию решают с помощью конструкции generic. В Verilog для этого используются параметры, то есть в описании модуля наряду с описанием переменных и сигналов могут описываться и параметры: 

module parity (bus, out); 
parameter width=8, typ_delay=1, modul_delay=typ_delay*2; 
wire [width-1:0] bus; 
wire out; 
assign #modul_delay out=^bus; 
endmodule 

Данный модуль описывает логику генерации бита четности и может применяться с различной шириной шины и различными задержками. При подключении модуля может сообщаться либо типовая задержка (по которой вычисляется задержка модуля), либо задержка модуля (в случае, когда моделируется нетипичная ситуация). 

Включать модуль в иерархию можно следующим образом: 

parity U1(bus1,out1); // будут использованы значения по умолчанию (width=8, typ_delay=1, modul_delay=typ_delay*2) 
parity #(4, 0.5) U2(bus2,out2); // ширина шины — 4 бита, типовая задержка 0.5 (задержка в модуле 0.5*2=1) 
parity #(16, 1, 5) U3(bus3,out3); // ширина шины — 16 бит, типовая задержка 1 (игнорируется), задержка в модуле 5

Иными словами, конструкция #(фактический параметр, фактический параметр, …) соответствует конструкции map generic языка VHDL. Может возникнуть путаница между конструкциями временного контроля и параметрами. Но, во-первых, временной контроль и привязка параметров происходят в разных конструкциях языка и по контексту видно значение знака #, а во-вторых, параметры, как правило, определяют временные задержки внутри модуля. Кроме такого механизма установки параметров существует механизм defparam и иерархические имена, позволяющие группировать все фактические параметры в модулях в одном месте вышестоящего модуля. Пример можно переписать следующим образом: 

module top; //модуль верхнего уровня 
… 
parity U1(bus1,out1); 
parity U2(bus2,out2); 
parity U3(bus3,out3); 
… 
defparam 
top.U2.width=4; 
top.U2.typ_delay=0.5; 
top.U3.width=16; 
top.U3.modul_delay=5; 
endmodule 

Так как при описании параметров могут употребляться выражения, данный механизм обеспечивает возможность гибкого повторного использования кода. Следует, однако, помнить, что не всякий синтез поддерживает иерархические имена и defparam, поэтому нужно пользоваться первым механизмом для написания синтезируемого кода. 

Препроцессор языка Verilog (директивы компилятора)

Наряду с механизмом параметров, описанным выше, в языке Verilog существует механизм, сходный с текстовым препроцессором языка С. Этот механизм (по аналогии с С назовем его препроцессором) позволяет проводить предварительную обработку текста до того, как данные будут обработаны симулятором или средством синтеза. Препроцессор обеспечивает условную компиляцию, выполнение макро- подстановок, интерфейс с внешней средой. Аналогом условной компиляции Verilog является конструкция if-generate языка VHDL. Для работы данного механизма, а также других директив компилятора используется символ апострофа «`» (не путать с кавычкой в литералах). Наиболее часто применяемые директивы препроцессора — `define, `include, `ifdef, `else, `endif. Таким же образом задаются директивы компилятора, не являющиеся директивами препроцессора, например `timescale. 

Для включения текста из одного файла в другой используется `include «Имя_Файла» (например, `include «dumppins.v»). Следует помнить, что модули Verilog не могут объявляться внутри модуля (не могут быть вложенными), поэтому `include можно использовать либо вне модуля, либо включать в модуль код, не содержащий модулей (module … endmodule).

Использование `define для описания текстового макроса можно проиллюстрировать следующим образом: 

`define nc @(negedge clk)

Далее в коде программы данный макрос может использоваться следующим образом: 

аlways `nc … или a<= `nc b.

Пример директив условной компиляции может быть следующим: 

`ifdef test0 
`define CPUTEST0 
`include «fault/test0» 
`else 
`define CPUTEST1 
`include «fault/test1» 
`endif 

Поскольку определить макрос можно из командной строки компилятора (опция +define+_NAME_), то данный механизм позволяет использовать конфигурационные скрипты. В настоящее время существуют средства синтеза как не поддерживающие препроцессор (старые или недоделанные версии), так и расширяющие стандартный набор директив. Например, средство синтеза может поддерживать директиву `for, что позволяет реализовать конструкцию for-generate языка VHDL. 

Конструкции, применяемые для симулирования библиотечных ячеек

Данная статья направлена на то, чтобы осветить поведенческое моделирование и написание такого поведенческого кода, который может быть синтезирован средством синтеза. Кроме этого, в языке существуют предопределенные модули, которые могут быть использованы как структурные элементы нижнего уровня. Однако некоторые их них практически никогда не используются, поскольку они неприменимы для задач разработки СБИС или ПЛИС. Это, например, описания транзисторных ключей: nmos, pmos, rnmos, rpmos, cmos, rcmos. 

Также существуют предопределенные логические ячейки: and, nand, nor, or, xor, xnor, buf, not, bufif0, bufif1, notif1, notif0. Они совпадают по названию с булевыми функциями и являются двухвходовыми элементами, выполняющими соответствующую функцию: not — инвертор, buf — буфер. Последние четыре элемента имеют вход разрешения и состояние высокого импеданса (z). Но так как любой из этих элементов может быть описан с использованием операторов языка, то смысл применения данных конструкций определяется предпочтением разработчика.Например, and U1(out,in1,in2) — то же самое, что out=in1&in2; bufif1 U2 (out, in, control) — то же самое, что out=control?in:1'bz.

Следует заметить, что для облегчения записи элементы могут подключаться на шину или иметь неименованные включения (instance), но все же использование операторов кажется предпочтительнее, а результат синтеза или поведение модели будут одинаковыми в обоих случаях. 

Для моделирования библиотечных элементов может применяться табличный механизм — UDP (User Defined Primitive). Это несинтезируемая конструкция, похожая по описанию на модуль, но подчиняющаяся более строгим правилам. Назначение UDP — моделировать логику, заданную таблицей истинности. При этом с помощью UDP можно описывать как комбинаторную, так и последовательную логику. UDP может иметь только один выход. 

primitive and_or(out, a1,a2,a3, b1,b2); 
output out; 
input a1,a2,a3, b1,b2; 
table 
//state table information goes here 
... 
endtable 
endprimitive 

Как видно, основным элементом UDP является таблица истинности. Ее элементы могут принимать следующие значения. 

	0
	Логический 0
	 

	1
	Логическая 1
	 

	х
	Не определено
	 

	?
	Любое значение из 0, 1 и x
	Не может использоваться в качестве выходного значения

	b
	Итерация 0 и 1
	Не может использоваться в качестве выходного значения

	-
	Без изменения
	Может использоваться на выходе только последовательного UDP

	(vw)
	Значение, изменяющееся от v до w
	v и w могут быть 0, 1, x,? или b

	*
	То же, что и ??
	Любое изменение значения на входе

	r
	То же, что и 01
	Прямой фронт на входе

	f
	То же, что и 10
	Обратный фронт на входе

	p
	Итерация (01), (0x) и (x1)
	Положительный фронт на входе

	n
	Итерация (10), (1x) и (x0)
	Отрицательный фронт на входе


Количество пробельных символов в описании таблицы не играет роли, важен порядок. Приведем пример комбинаторного UDP (мультиплексора): 

primitive multiplexer(mux,control,dataA,dataB ); 
output mux; 
input control, dataA, dataB; 
table 
// control dataA dataB mux 
0 1?: 1; //? = 0,1,x 
0 0?:0; 
1?1:1; 
1?0:0; 
x 0 0:0; 
x 1 1:1; 
endtable 
endprimitive

Пример последовательного UDP: 

primitive srff (q,s,r); 
output q; 
input s,r; 
reg q; 
initial q = 1'b1; // initial statement specifies that output 
// terminal q has a value of 1 at the start 
// of the simulation 
table 
// s r q q+ 
1 0:?: 1; 
f 0: 1: -; 
0 r:?: 0; 
0 f: 0: -; 
1 1:?: 0; 
endtable 
endprimitive 

Синтезируемое подмножество языка

Прежде чем вести разговор о синтезируемом подмножестве языка, следует остановиться на общих принципах разработки СБИС или ПЛИС с использованием языков высокого уровня. Производитель микросхемы ПЛИС или фабрика, производящая СБИС, предоставляет модель библиотечных элементов, которые могут быть использованы в схеме. Эти библиотеки предназначены для использования в различных областях проектирования (моделирование, синтез, топология кристалла и т. д.) и могут быть выполнены в различных форматах. Один из форматов представляет собой библиотеку элементов, описанных на языке высокого уровня и предназначенных для моделирования. В данной библиотеке содержатся элементы, описанные посредством несинтезируемых конструкций. В описании библиотечных элементов встречаются не рассмотренные в статье конструкции для описания сквозных задержек распространения (path delay) и механизмы контроля временных параметров (setup, hold time). Сквозные задержки позволяют описать задержку модуля, не вдаваясь в его внутреннюю структуру, и поддерживаются специальными словами языка (specify, specparam, endspecify) и специальным синтаксисом (например, (a,b,cґ>x,y,z) = Tin; (d+=>x) = Tout;). А системные функции контроля используются для проверки того, происходят ли изменения сигналов в нужные моменты времени, например, для того, чтобы последовательная логика не попадала в метастабильное состояние. Мы не останавливались подробно на этих элементах, поскольку пользователю средств синтеза не нужно описывать библиотечные элементы, а в текстовом описании библиотеки обычно содержится информация о поведении элемента и его временной диаграмме. Можно сказать, что при разработке проекта СБИС или ПЛИС, возможно, даже не придется вручную подключать/отключать эти элементы в структурном описании и вообще иметь сведения об их существовании. 

Следующие два элемента разработки предоставляются пользователем — это синтезируемое описание на языке HDL и набор директив для средства синтеза. Также нужен набор средств разработки, включающий в себя симулятор и средство синтеза. Следует остановиться на отличиях синтеза от компиляторов поведенческих языков. Компилятор языка (например С) детерминированным образом пере- водит операторы языка в команды машинного языка. Синтез переводит последовательные операторы языка HDL в структурную схему, состоящую из библиотечных элементов. Данный процесс больше похож на перебор вариантов и выбор наилучшего, удовлетворяющего временным ограничениям и занимаемой площади. Следствием этого является то, что синтез — итеративный процесс, в котором результаты предыдущего шага используются для коррекции директив для следующего шага. Методология написания директив синтеза и подход к синтезу схемы, имеющей минимальное количество вентилей и максимальную тактовую частоту, является отдельным вопросом и в данной статье не рассматривается. Но в любом случае предоставляемое пользователем описание на языке HDL должно быть правильным — работоспособным и синтезируемым.

Рассмотрим операции, которые необходимо выполнить для разработки прибора на СБИС или ПЛИС: 
1) подготовка синтезируемого поведенческого описания схемы (этому шагу может предшествовать моделирование на С, в среде MatLab или с использованием специализированных средств); 
2) написание тестовой оболочки — испытательного стенда (testbench), — в которой проводится полное тестирование всех режимов модели; 
3) итеративная процедура синтеза поведенческой модели, результатом которой является структурная схема (netlist) с использованием библиотечных элементов и файл задержек распространения (standard delay file, SDF); 
4) проверка работоспособности netlist. При этом обычно используется тот же testbench, что и для п. 2. Нарушение работоспособности может быть связано с нарушениями допустимых времен библиотечных элементов, гонками фронтов и пр. В зависимости от опыта разработчика происходит возврат к п. п. 1–3 или переход к п. 5; полезно также на этом шаге проверить результаты работы средств временного анализа (static timing report) для выявления «узких мест» — критических путей проекта; 
5) процедура размещения и трассировки — выполняется специальными средствами; результатом является коррекция SDF-файла (помимо прошивки ПЛИС или разводки СБИС); 
6) проверка netlist с новыми задержками; подобна п. 4, но возврат возможен к п. п. 1, 3, 5; 
7) in-place оптимизация (для большинства современных ПЛИС отсутствует), в ходе которой производится выравнивание времен распространения сигналов — усиление/ослабление выходов ячеек, установка буферов/элементов задержки; 
8) повтор п. 6 до тех пор, пока не будут достигнуты требуемые характеристики; 
9) подготовка производственных тестов, предназначенных для поиска неисправностей в СБИС (для ПЛИС этот шаг необязателен). 

Шаг 3 в данной схеме раньше выполнялся вручную, когда поведенческое описание заменялось на структурное самим разработчиком. В настоящее время существуют средства синтеза, которые позволяют получить более эффективную схему. Пользование этими средствами накладывает ограничение на употребление конструкций языка в исходном коде. Требуется синтезируемая модель. Кроме того, чтобы многократно не переписывать поведенческую модель, следует четко представлять, какое Verilog-описание приведет к синтезу того или иного элемента схемы.

Ранее в статье было продемонстрировано, как работает Verilog-симулятор. Далее будут рассмотрены требования средств синтеза. 

В результате синтеза любой элемент языка может быть синтезирован, проигнорирован, либо может вызвать ошибку. Конструкции языка могут поддерживаться полностью, частично или не поддерживаться вовсе.

Как уже упоминалось, иерархические имена, initial, fork-join или primitive, не поддерживаются. Временной контроль #nn игнорируется в синтезируемой модели. Событийный контроль поддерживается частично: только в блоках always. 

В документации к средству синтеза указывается, какие ограничения вводятся на элементы языка. Перед написанием модели следует ознакомиться с документацией.Кроме этого, существуют правила описания комбинаторной логики, последовательной логики, мультиплексоров, FSM и т. п.

Комбинаторная логика синтезируется из следующих конструкций: 
1) непрерывное присвоение: 

assign a=b+c&d; 
wire b={e,f} | g[1:0]; 

2) сигналы, описанные в функциях; 
3) сигналы, описанные следующей или подобной конструкцией: 

reg data_out; 
always @(a or b or c) 
if (b) 
data_out = a; 
else 
data_out = c; 

то есть блок always, в списке чувствительности которого перечислены все входные сигналы. 
Как можно видеть в третьем (а иногда и во втором) случае, объявление сигнала с ключевым словом reg не приводит к созданию регистра. Также комбинаторная логика синтезируется в случае case, если описаны все ветви. При этом получается не приоритетный набор (как в моделировании), а набор параллельных конструкций. Также case используется для синтеза мультиплексоров. 

Последовательная логика имеет ограничения в синтезируемых конструкциях. Для описания регистра-защелки (latch) применяется следующая конструкция: 

reg data_out; 
always @(data_in or enable) 
if (enable) 
data_out = data_in; 

Для регистров, работающих по фронту/срезу сигнала, применяется описание с конструкциями posedge или negedge: 

reg data_out; 
always @(posedge clock) 
data_out = data_in; 

Чтобы добавить синхронный сброс, описание нужно дополнить сигналом установки/сброса, но не вносить эти сигналы в список чувствительности: 

reg data_out; 
always @(posedge clock) 
if (set_sig) 
data_out = 1'b1; 
else if (reset_sig) 
data_out = 1'b0; 
else 
data_out = data_in; 

Для асинхронной установки/сброса конструкция должна быть изменена, так чтобы эти сигналы попали в список чувствительности: 

reg data_out; 
always @(posedge clock or posedge set_sig or posedge reset_sig) 
if (set_sig) 
data_out = 1'b1; 
else if (reset_sig) 
data_out = 1'b0; 
else 
data_out = data_in; 

Пользуясь этими принципами, можно описать триггер любого типа.

Если сигнал имеет размерность, большую единицы, то синтезируются устройства для каждого бита, а существующая логика или арифметика в подобных конструкциях синтезируется в комбинаторную схему. 

Блок case, у которого расшифровываются не все входные воздействия, синтезируется в элемент последовательной логики. Также может использоваться для описания FSM (finite state machine). Поэтому кажется более правильным воспользоваться комбинаторным case и явно описать регистр для хранения результата (к примеру, для реализации FSM).

Ограничения, накладываемые средством синтеза на язык HDL, позволяют описывать логику работы схемы в виде комбинаторной логики и регистров для хранения результатов. Такое описание называется RTL (register transfer level) и иногда отделяется (по смыслу) от поведенческого. Основная идея, которую хотелось бы высказать в данной статье, — это то, что следует пользоваться RTL-описанием для моделирования и синтеза. Данный подход является эффективным методом ведения разработки. Так как «правильное» (написанное с применением правил, изложенных выше) поведенческое описание может быть автоматически синтезировано на уровень вентилей, это позволяет значительно ускорить и упростить процесс разработки. При этом, если временные ограничения были выбраны правильно (с учетом задержек в элементах библиотеки), поведение RTL (поведенческой модели) не будет отличаться от поведения нетлиста (gate-level) и не должно отличаться от поведения готового изделия. 

Завершая рассмотрение синтезируемого (RTL) кода, следует упомянуть о директивах синтеза, не являющихся элементами языка Verilog. Они задаются в поле комментария и игнорируются при моделировании, но управляют средством синтеза, например: 

// synopsis synthesis off 
$write(«this string is ignored by synthesis»); 
// synopsis synthesis on 

Для уже упоминавшегося case существует директива синтеза (выбирающая синтез в последовательную или параллельную логику), которая может быть указана в поле комментария: 

// ambit synthesis case = full | parallel | mux 

Однако данный механизм нестандартен и его использование оправдано в редких случаях. 

Заключение

Данная статья не является документацией языка Verilog. Целью ее написания было проиллюстрировать работу Verilog-симулятора и показать возможность использования языка для разработки цифровых схем. При этом рассматривались как аспекты, связанные с тестированием модели (написание испытательных стендов), так и проблемы написания синтезируемого кода. Приведенные примеры и объяснения к ним должны были (по замыслу автора) показать различия между процедурными языками и языками описания HDL. Развитие CSoC и ПЛИС позволяет предположить, что все большие части проекта могут быть реализованы в виде описания HDL. 

В статье была сделана попытка показать, что Verilog является сильным и удобным средством для достижения этой цели

